CHAPTER 16

Computer Algebra with Formulas (15.9)–(15.18)

The research presented in [**19, 20, 21**] was made possible when (15.16) was discovered and systems of computer algebra could then be used to find several key identities through trial-and-error experimentation. Similarly, one can make interesting discoveries or rediscoveries merely by using the formulas for $c_i^*(z)$ and $c_i^{**}(\zeta)$ with a few basic commands in a system of computer algebra. Here, we illustrate how that can be done by selecting a version of *Mathematica* from [**55, 56, 57, 58, 59**] as the system. The names of its commands indicate well what they do.

16.1. Computer representations for $c_i^*(z)$ and $c_i^{**}(\zeta)$

We apply (15.9), (15.12), (15.17), (15.18), and (15.16) with the selected version of *Mathematica* to conclude that successive notebook evaluations of

```
c[m_,0][z_] := 1

cS[m_,i_][z_] := Sum[Binomial[m-j,i-j]*
  (D[rho[z],{z,i-j}]/rho[z])*c[m,j][z],{j,0,i}]

alpha[0,j_][zeta_] := 1

alpha[i_,j_][zeta_] := ( Sum[alpha[i-1,k]'[zeta]
  -(i-1+k)(f''[zeta]/f'[zeta])*
  alpha[i-1,k][zeta],{k,1,j}] ) /; i >= 1

cSS[m_,i_][zeta_] := Sum[alpha[i-j,m-i][zeta]*
  (f'[zeta])^j*c[m, j][f[zeta]],{j,0,i}]
```

enable *Mathematica* to then give computer representations for $c_i^*(z)$ and $c_i^{**}(\zeta)$, when $i = 0, 1, 2, \ldots$ and m can remain a symbol for any positive integer $\geq i$. For instance, the computer representations for the evaluations of `cS[m,1][z]` and `cSS[m,1][zeta]` show that $c_1^*(z)$ and $c_1^{**}(\zeta)$ are respectively given by

$$c_1^*(z) \equiv c_1(z) + m\frac{\rho'(z)}{\rho(z)} \quad \text{and} \quad c_1^{**}(\zeta) \equiv f'(\zeta)\, c_1\big(f(\zeta)\big) - \binom{m}{2}\frac{f''(\zeta)}{f'(\zeta)}.$$

Also, the computer representation for the evaluation of `cS[m,2][z]` yields

$$c_2^*(z) \equiv c_2(z) + (m-1)\, c_1(z)\, \frac{\rho'(z)}{\rho(z)} + \binom{m}{2}\frac{\rho''(z)}{\rho(z)}.$$

16.2. Applications based on the representations for $c_i^*(z)$ and $c_i^{**}(\zeta)$

EXAMPLE 16.1. With $m \geq 2$ and symbols r_1, r_2 for rational numbers, we set

$$\boldsymbol{P}_{m,2} \equiv \boldsymbol{w}_2^{(0)} + r_1 \big(\boldsymbol{w}_1^{(0)}\big)^2 + r_2 \, \boldsymbol{w}_1^{(1)}. \tag{16.1}$$

In regard to the function $P_{m,2}(z)$ on Ω that is obtained by replacing each $\boldsymbol{w}_i^{(j)}$ in $\boldsymbol{P}_{m,2}$ with the corresponding $c_i^{(j)}(z)$ from (15.9), we see that the evaluation of

```
P[z_] := c[m,2][z] + r1*c[m,1][z]^2 + r2*c[m,1]'[z]
```

represents $P_{m,2}(z)$. Also, for the function $P_{m,2}^*(z)$ on Ω that is obtained by replacing each $\boldsymbol{w}_i^{(j)}$ in $\boldsymbol{P}_{m,2}$ with the corresponding $c_i^{*(j)}(z)$ from (15.12), the evaluation of

```
PS[z_] := cS[m,2][z] + r1*cS[m,1][z]^2 + r2*cS[m,1]'[z]
```

represents $P_{m,2}^*(z)$. There are eight terms in the output for the evaluation of

```
dif1[z_] = Expand[ PS[z] - P[z] ]
```

and in those terms the parts not involving `m`, `r1`, `r2` are equal to the evaluations of

```
b[1] = c[m,1][z]*rho'[z]/rho[z];

b[2] = (rho'[z]/rho[z])^2;

b[3] = rho''[z]/rho[z];
```

while the evaluations of

```
a[1] = Coefficient[dif1[z],b[1]];

a[2] = Coefficient[dif1[z],b[2]];

a[3] = Coefficient[dif1[z],b[3]];
```

then yield the respective coefficients `a[1]`, `a[2]`, `a[3]` of `b[1]`, `b[2]`, `b[3]` in `dif1[z]`. Of course, if `r1`$= r_1$ and `r2`$= r_2$ are specific rational numbers, then we see that: `a[1]`, `a[2]`, `a[3]` are zero if and only if `PS(z) - P(z)` is zero and $P_{m,2}^*(z) \equiv P_{m,2}(z)$. After the evaluation of

```
list1 = {a[1]==0, a[2]==0, a[3]==0}
```

as a system of three linear equations in `r1` and `r2`, the evaluation of

```
Solve[list1, {r1,r2}]
```

yields a unique solution that corresponds to

$$r_1 \equiv -\frac{(m-1)}{2m} \quad \text{and} \quad r_2 \equiv -\frac{(m-1)}{2}. \tag{16.2}$$

Thus, when r_1, r_2 for (16.1) are defined by (16.2), we have $P_{m,2}^*(z) \equiv P_{m,2}(z)$ on Ω as a valid identity for any (15.9) on Ω having $m \geq 2$ and any transformation (15.10) of that (15.9) into a corresponding equation (15.11) on Ω.

EXAMPLE 16.2. With $m \geq 2$ and symbols s_1, s_2 for rational numbers, we set

$$\boldsymbol{Q}_{m,2} \equiv \boldsymbol{w}_2^{(0)} + s_1 \big(\boldsymbol{w}_1^{(0)}\big)^2 + s_2\, \boldsymbol{w}_1^{(1)}. \tag{16.3}$$

In regard to the function $Q_{m,2}(z)$ on Ω that is obtained by replacing each $\boldsymbol{w}_i^{(j)}$ in $\boldsymbol{Q}_{m,2}$ with the corresponding $c_i^{(j)}(z)$ from (15.9), we see that the evaluation of

```
Q[z_] := c[m,2][z] + s1*c[m,1][z]^2 + s2*c[m,1]'[z]
```

represents $Q_{m,2}(z)$. For the function $Q^{**}_{m,2}(\zeta)$ on Ω^{**} that is obtained by replacing each $\boldsymbol{w}_i^{(j)}$ in $\boldsymbol{Q}_{m,2}$ with the corresponding $c_i^{**(j)}(\zeta)$ from (15.16), the evaluation of

```
QSS[zeta_] := ( cSS[m,2][zeta] + s1*cSS[m,1][zeta]^2
                + s2*cSS[m,1]'[zeta] )
```

represents $Q^{**}_{m,2}(\zeta)$. There are twenty terms in the output for the evaluation of

```
dif2[zeta_] = Expand[ QSS[zeta] - (f'[zeta])^2*Q[f[zeta]] ]
```

and in those terms the parts not involving `m`, `s1`, `s2` are given by the evaluations of

```
b[4] = c[m, 1][f[zeta]] f''[zeta];

b[5] = (f''[zeta]/f'[zeta])^2;

b[6] = f'''[zeta]/f'[zeta];
```

while the evaluations of

```
a[4] = Coefficient[dif2[zeta],b[4]];

a[5] = Coefficient[dif2[zeta],b[5]];

a[6] = Coefficient[dif2[zeta],b[6]];
```

give the coefficients of `b[4]`, `b[5]`, `b[6]` in `dif2[zeta]`. Naturally, if `s1`$= s_1$ and `s2`$= s_2$ are specific rational numbers, then we see that: `a[4]`, `a[5]`, `a[6]` are zero if and only if `QSS[zeta] - (f'[zeta])^2*Q[f[zeta]]` is zero and we have the identity $Q^{**}_{m,2}(\zeta) \equiv \big(f'(\zeta)\big)^2\, Q_{m,2}\big(f(\zeta)\big)$. After the evaluation of

```
list2 = {a[4]==0, a[5]==0, a[6]==0}
```

as a system of three linear equations in `s1` and `s2`, the evaluation of

```
Solve[list2, {s1,s2}]
```

yields a unique solution that corresponds to

$$s_1 \equiv -\frac{(m-2)(3m-1)}{6m(m-1)} \quad \text{and} \quad s_2 \equiv -\frac{m-2}{3}. \tag{16.4}$$

Thus, for s_1, s_2 in (16.3) defined by (16.4), we have $Q^{**}_{m,2}(\zeta) \equiv \big(f'(\zeta)\big)^2\, Q_{m,2}\big(f(\zeta)\big)$ on Ω^{**} as a valid identity for any equation (15.9) on Ω having $m \geq 2$ and any transformation (15.14) of that (15.9) into a corresponding equation (15.15) on Ω^{**}.

EXAMPLE 16.3. Here, we use the computer representations for $c_i^*(z)$ and $c_i^{**}(\zeta)$ in Section 16.1 to check that the expression for $\mathcal{I}_{4,1;\,4}$ in (1.17) on page 4 is printed correctly. We find that the evaluation of

```
Simplify[ ( cS[4,4][z] -(1/4)cS[4,1][z]*cS[4,3][z]
  -(1/2)cS[4,3]'[z] -(9/100)cS[4,2][z]^2
  +(1/5)cS[4,2]''[z] +(13/100)cS[4,1][z]^2*cS[4,2][z]
  +(27/100)cS[4,1]'[z]*cS[4,2][z] +(1/4)cS[4,1][z]*cS[4,2]'[z]
  -(39/1600)cS[4,1][z]^4 -(39/200)cS[4,1][z]^2*cS[4,1]'[z]
  -(33/200)(cS[4,1]'[z])^2 -(3/20)cS[4,1][z]*cS[4,1]''[z]
  -(1/20)cS[4,1]'''[z] )
       - ( c[4,4][z] -(1/4)c[4,1][z]*c[4,3][z]
  -(1/2)c[4,3]'[z] -(9/100)c[4,2][z]^2
  +(1/5)c[4,2]''[z] +(13/100)c[4,1][z]^2*c[4,2][z]
  +(27/100)c[4,1]'[z]*c[4,2][z] +(1/4)c[4,1][z]*c[4,2]'[z]
  -(39/1600)c[4,1][z]^4 -(39/200)c[4,1][z]^2*c[4,1]'[z]
  -(33/200)(c[4,1]'[z])^2 -(3/20)c[4,1][z]*c[4,1]''[z]
  -(1/20)c[4,1]'''[z] ) ]
```

is zero and the evaluation of

```
Simplify[ ( cSS[4,4][zeta]
  -(1/4)cSS[4,1][zeta]*cSS[4,3][zeta]
  -(1/2)cSS[4,3]'[zeta] -(9/100)cSS[4,2][zeta]^2
  +(1/5)cSS[4,2]''[zeta]
  +(13/100)cSS[4,1][zeta]^2*cSS[4,2][zeta]
  +(27/100)cSS[4,1]'[zeta]*cSS[4,2][zeta]
  +(1/4)cSS[4,1][zeta]*cSS[4,2]'[zeta]
  -(39/1600)cSS[4,1][zeta]^4
  -(39/200)cSS[4,1][zeta]^2*cSS[4,1]'[zeta]
  -(33/200)(cSS[4,1]'[zeta])^2
  -(3/20)cSS[4,1][zeta]*cSS[4,1]''[zeta]
  -(1/20)cSS[4,1]'''[zeta] )
       - (f'[zeta])^4( c[4,4][f[zeta]]
  -(1/4)c[4,1][f[zeta]]*c[4,3][f[zeta]]
  -(1/2)c[4,3]'[f[zeta]] -(9/100)c[4,2][f[zeta]]^2
  +(1/5)c[4,2]''[f[zeta]]
  +(13/100)c[4,1][f[zeta]]^2*c[4,2][f[zeta]]
  +(27/100)c[4,1]'[f[zeta]]*c[4,2][f[zeta]]
  +(1/4)c[4,1][f[zeta]]*c[4,2]'[f[zeta]]
  -(39/1600)c[4,1][f[zeta]]^4
  -(39/200)c[4,1][f[zeta]]^2*c[4,1]'[f[zeta]]
  -(33/200)(c[4,1]'[f[zeta]])^2
  -(3/20)c[4,1][f[zeta]]*c[4,1]''[f[zeta]]
  -(1/20)c[4,1]'''[f[zeta]] ) ]
```

is zero. Consequently, $\mathcal{I}_{4,1;\,4}$ as presented in (1.17) on page 4 is a relative invariant of weight $s = 4$ for the equations (15.9) on page 158 having order $m = 4$.

EXAMPLE 16.4. With $m \geq 3$ and symbols t_1, t_2, t_3, t_4, t_5 representing rational numbers, we introduce

$$\boldsymbol{I}_{m,3} \equiv \boldsymbol{w}_3 + t_1\,\boldsymbol{w}_1\,\boldsymbol{w}_2 + t_2\,\big(\boldsymbol{w}_1\big)^3 + t_3\,\boldsymbol{w}_2^{(1)} + t_4\,\boldsymbol{w}_1\,\boldsymbol{w}_1^{(1)} + t_5\,\boldsymbol{w}_1^{(2)}. \tag{16.5}$$

For the function $I_{m,3}(z)$ on Ω that is obtained by replacing each $\boldsymbol{w}_i^{(j)}$ in $\boldsymbol{I}_{m,3}$ with the corresponding $c_i^{(j)}(z)$ from (15.9), the evaluation of

```
Inv[z_] := ( c[m,3][z] + t1*c[m,1][z]*c[m,2][z]
             + t2*c[m,1][z]^3 + t3*c[m,2]'[z]
             + t4*c[m,1][z]*c[m,1]'[z] + t5*c[m,1]''[z] )
```

represents $I_{m,3}(z)$. For the function $I^*_{m,3}(z)$ on Ω that is obtained by replacing each $\boldsymbol{w}_i^{(j)}$ in $\boldsymbol{I}_{m,3}$ with the corresponding $c_i^{*(j)}(z)$ from (15.12), the evaluation of

```
InvS[z_] := ( cS[m,3][z] + t1*cS[m,1][z]*cS[m,2][z]
  + t2*cS[m,1][z]^3 + t3*cS[m,2]'[z]
  + t4*cS[m,1][z]*cS[m,1]'[z] + t5*cS[m,1]''[z] )
```

represents $I^*_{m,3}(z)$. For the function $I^{**}_{m,3}(\zeta)$ on Ω^{**} that is obtained by replacing each $\boldsymbol{w}_i^{(j)}$ in $\boldsymbol{I}_{m,3}$ with the corresponding $c_i^{**(j)}(\zeta)$ from (15.16), the evaluation of

```
InvSS[zeta_] := ( cSS[m,3][zeta]
  + t1*cSS[m,1][zeta]*cSS[m,2][zeta] + t2*cSS[m,1][zeta]^3
  + t3*cSS[m,2]'[zeta] + t4*cSS[m,1][zeta]*cSS[m,1]'[zeta]
  + t5*cSS[m,1]''[zeta] )
```

represents $I^{**}_{m,3}(\zeta)$. We note that t_1, t_2, t_3, t_4, t_5 for (16.5) yield

$$I^*_{m,3}(z) \equiv I_{m,3}(z) \text{ on } \Omega, \text{ and } I^{**}_{m,3}(\zeta) \equiv \big(f'(\zeta)\big)^3\, I_{m,3}\big(f(\zeta)\big), \text{ on } \Omega^{**}. \tag{16.6}$$

if and only if their representations `t1`, `t2`, `t3`, `t4`, `t5` make the evaluations of

```
diff1[z_] = Expand[ InvS[z] -  Inv[z] ]

diff2[zeta_] = Expand[InvSS[zeta]-(f'[zeta])^3*Inv[f[zeta]]]
```

identically zero. Among the thirty-eight terms in the expansion of `diff1[z]`, there are eight parts that do not involve `m`, `t1`, `t2`, `t3`, `t4`, `t5`. Let them be copied individually from the output, pasted into individual input cells, given the names `b3[1]`, `b3[2]`, ..., `b3[8]`, and then evaluated. Among the ninety-three terms in the expansion of `diff2[zeta]`, there are eight parts that do not involve `m`, `t1`, `t2`, `t3`, `t4`, `t5`. Let them be copied from the output, pasted into input cells, given the names `b3[9]`, `b3[10]`, ..., `b3[16]`, and then be evaluated. We evaluate

```
Do[a3[k] = Coefficient[diff1[z], b3[k]], {k,1,8}];

Do[a3[k] = Coefficient[diff2[zeta], b3[k]], {k,9,16}];
```

and then find that the evaluation of

```
Solve[ Table[a3[k] == 0, {k,1,16}], {t1,t2,t3,t4,t5} ]
```

yields a unique solution. As expressed for (16.5), it is given by

$$t_1 = -\frac{m-2}{m}, \qquad t_2 = \frac{(m-1)(m-2)}{3m^2}, \qquad t_3 = -\frac{m-2}{2}, \tag{16.7}$$
$$t_4 = \frac{(m-1)(m-2)}{2m}, \qquad \text{and} \qquad t_5 = \frac{(m-1)(m-2)}{12}.$$

Thus, (16.6) is satisfied by (16.5) with (16.7) for each equation (15.9) having $m \geq 3$ as well as each transformation (15.10) of (15.9) into a corresponding (15.11) and each transformation (15.14) of (15.9) into a corresponding (15.15). In this regard, see (1.13) of page 3. If the definitions of `b3[1]`, `b3[2]`, ..., `b3[16]` give difficulty, use the Google browser *Chrome* to visit

`http://homepages.uc.edu/~chalklr/Chapter-16.html`

and then download the *Mathematica* notebook available there. Details are also given in that notebook for Examples 16.1, 16.2, 16.3, and 16.5.

EXAMPLE 16.5. There are unique rational numbers $u_1, u_2, \ldots, u_{12}$ for

$$\begin{aligned} \boldsymbol{I}_{m,4} \equiv \boldsymbol{w}_4 &+ u_1\, \boldsymbol{w}_1 \boldsymbol{w}_3 + u_2\, \boldsymbol{w}_3^{(1)} + u_3 \left(\boldsymbol{w}_2\right)^2 + u_4\, \boldsymbol{w}_2^{(2)} + u_5 \left(\boldsymbol{w}_1\right)^2 \boldsymbol{w}_2 \\ &+ u_6\, \boldsymbol{w}_1^{(1)} \boldsymbol{w}_2 + u_7\, \boldsymbol{w}_1 \boldsymbol{w}_2^{(1)} + u_8 \left(\boldsymbol{w}_1\right)^4 + u_9 \left(\boldsymbol{w}_1\right)^2 \boldsymbol{w}_1^{(1)} \\ &+ u_{10} \big(\boldsymbol{w}_1^{(1)}\big)^2 + u_{11}\, \boldsymbol{w}_1 \boldsymbol{w}_1^{(2)} + u_{12}\, \boldsymbol{w}_1^{(3)}, \quad \text{with } m \geq 4, \end{aligned} \tag{16.8}$$

such that the functions $I_{m,4}(z)$ on Ω, $I^*_{m,4}(z)$ on Ω, and $I^{**}_{m,4}(\zeta)$ on Ω^{**} that are obtained by replacing each $\boldsymbol{w}_i^{(j)}$ in $\boldsymbol{I}_{m,4}$ with the corresponding $c_i^{(j)}(z)$ from (15.9), with the $c_i^{*(j)}(z)$ from (15.11), and with the $c_i^{**(j)}(\zeta)$ from (15.15), satisfy both

$$I^*_{m,4}(z) \equiv I_{m,4}(z) \quad \text{on } \Omega, \text{ and} \quad I^{**}_{m,4}(\zeta) \equiv \big(f'(\zeta)\big)^4\, I_{m,4}\big(f(\zeta)\big), \quad \text{on } \Omega^{**}.$$

When the technique of Example 4.4 is repeated here, the main difference is that: in place of the copy and paste for Example 4.4 where `b3[k]` was obtained separately for $1 \leq k \leq 8$ and $9 \leq k \leq 16$, we now use copy and paste to obtain `b4[k]` separately for $1 \leq k \leq 20$ and for $21 \leq k \leq 40$. Of course, this requires more patience. However, when details similar to those of Example 4.4 are carried out, the coefficients for $\boldsymbol{I}_{m,4}$ in (16.8) are found to be

$$\begin{aligned} & u_1 = -\tfrac{m-3}{m}, \quad u_2 = -\tfrac{m-3}{2}, \quad u_3 = -\tfrac{(m-2)(m-3)(5m+7)}{10(m+1)(m)(m-1)}, \\ & u_4 = \tfrac{(m-2)(m-3)}{10}, \quad u_5 = \tfrac{(m-2)(m-3)(5m+6)}{5(m+1)m^2}, \quad u_6 = \tfrac{(m-2)(m-3)(5m+7)}{10(m+1)m}, \\ & u_7 = \tfrac{(m-2)(m-3)}{2m}, \quad u_8 = -\tfrac{(m-1)(m-2)(m-3)(5m+6)}{20(m+1)m^3}, \\ & u_9 = -\tfrac{(m-1)(m-2)(m-3)(5m+6)}{10(m+1)m^2}, \quad u_{10} = -\tfrac{(m-1)(m-2)(m-3)(2m+3)}{20(m+1)m}, \\ & u_{11} = -\tfrac{(m-1)(m-2)(m-3)}{10m}, \quad u_{12} = -\tfrac{(m-1)(m-2)(m-3)}{120}. \end{aligned} \tag{16.9}$$

By setting $m = 4$ in these formulas, we obtain the coefficients for (1.17) on page 4.

Observation. The basic relative invariants $\boldsymbol{\mathcal{I}}_{m,1;\,s}$ of weight $s \geq 3$ for the equations (15.9) of order $m \geq s$ are given explicitly by the computer program in Section 6.1 on pages 53–54. We note that $\boldsymbol{I}_{m,3}$ in (16.5) with the coefficients of (16.7) is equal to $\boldsymbol{\mathcal{I}}_{m,1;\,3}$. Also, $\boldsymbol{I}_{m,4}$ in (16.8) with the coefficients of (16.9) is equal to $\boldsymbol{\mathcal{I}}_{m,1;\,4}$.